

A Brief Manual for LINMOD Version 3.4

David H. Christiansen, James D. Hosking, Ronald W. Helms,
Keith E. Muller, and Kevin B. Hunter

LINMOD (LINear MODels) performs a wide variety of General Linear Multivariate Model (GLMM) computations in SAS/IML. The software is especially useful for repeated measures, complex designs, and teaching. The source code (IML with a tiny amount of macro language), installation instructions, an extensive manual, and sample programs are available at <http://ehpr.ufl.edu/muller>. No warranty is provided or implied. An example follows.

```
TITLE1 "EXAMPLE0SAS, use LINMOD";
%LET LMDIRECT = ..\SOURCE\ ;
%INCLUDE "&LMDIRECT.MACROLIB.SAS" / NOSOURCE2 ;
&PROCSSCP DATA=PAYNE ;
VAR CONTROL LOW MODERATE HIGH INIT SCORE2 SCORE4 SCORE6 SCORE8 SCORE10;
PROC IML WORKSIZE=1000 SYMSIZE=1000; *4.1 Start IML;
&LINMOD ; *4.2 Grab code;
RUN GETCORSS; * 5. Retrieve the file _CORRDS_ created in Step 3 ;
* 6. Define the model and estimate primary parameters;
INDVARS = { CONTROL LOW MODERATE HIGH INIT };
DEPVAR=NAMELIST("SCORE",2,10,2);
RUN FITMODEL;
* 7.1 Conduct a test (and estimation) step;
PRINT  "* MANOVA Test of Main Effect of Treatment,  *" ;
C = { 1 -1 0 0 0 ,
      1 0 -1 0 0 ,
      1 0 0 -1 0 } ;
THETARNM= { "C - LOW" "C - MOD" "C - HIGH" };
*U defaults to Identity matrix, if NROW(U)=0;
RUN TESTGLH;
```

LINMOD Options and Default Values

SETOPT/General	MAKISS	GETCORSS	FITMODEL	TESTGLH
AVAILOPT	CHKMISS (ON)	CPARMS	PARMIN	C (ON)
LISTINFO	MPARMS	CSS	SSIN	U (ON)
CURROPTS	MSS		BETA (ON)	THETA0 (ON)
NEWOPTS			XPXINV	THETA (ON)
COMPRESS (ON)			UNIBETA	MID
NOPRINT			CHECK	EXTHETA (ON)
			EXBETA (ON)	MATTHETA
			COVBETA	UNITHETA (ON)
			SIGMA (ON)	MSH
			SCORR (ON)	MSE
			SSSTEP	ECORR (ON)
			SSFIT	HEIVAL
			LTFR	CANVEC
			LINDEP (ON)	CANRSQ (ON)
			PARMOUT (ON)	MULTTEST (ON)
				RSQUARED (ON)
				UNIREP (ON)
				UNIRPRNT (ON)
				UNIRWARN
				UNIRFORC
				UNIBOTH (ON)
				UNIRANK

(ON) indicates the default ON. The absence of (ON) indicates a default OFF.

```

TITLE1 "EXAMPLE1.SAS--Demonstrate simple LINMOD use";
* 0. Define raw data file;
FILENAME IN01 "..\EXAMPLES\PAYNE.DAT";
DATA PAYNE;      INFILE IN01;
* 1. Define directory in which LINMOD source code stored;

  * Change highlighted text in next line for your computer;
%LET LMDIRECT = ..\SOURCE\ ;
* 2. Define SAS macro code needed;
%INCLUDE "&LMDIRECT.MACROLIB.SAS" / NOSOURCE2 ;
* 3. Reduce raw data to a TYPE=CORR file named _CORRDS_ ;
&PROCSSCP DATA=PAYNE ;
VAR CONTROL LOW MODERATE HIGH INIT
    SCORE2 SCORE4 SCORE6 SCORE8 SCORE10;
PROC IML WORKSIZE=1000 SYMSIZE=1000; *4.1 Start IML;
&LINMOD ;                          *4.2 Grab code;
* 5. Retrieve the file _CORRDS_ created in Step 3 ;
RUN GETCORSS;
* 6. Define the model and estimate primary parameters;
INDVARS = { CONTROL LOW MODERATE HIGH INIT };
DEPVAR=NAMELIST("SCORE",2,10,2);
RUN FITMODEL;
* 7.1 Conduct a test (and estimation) step;
PRINT  "** MANOVA Test of Main Effect of Treatment,  *"
      , "** comparing each treatment to control group  *" ;
C = { 1 -1  0  0  0 ,
      1  0 -1  0  0 ,
      1  0  0 -1  0 } ;
THETARNM= { "C - LOW"  "C - MOD"  "C - HIGH" };
*U defaults to Identity matrix, if NROW(U)=0;
RUN TESTGLH;

```

TITLE1 "EXAMPLE2.SAS--Demonstrate LINMOD using MAKES";

```

FILENAME IN01 "..\EXAMPLES\PAYNE.DAT";
DATA PAYNE;
    INFILE IN01;
%LET LMDIRECT = ..\SOURCE\ ;
%INCLUDE "&LMDIRECT.MACROLIB.SAS" / NOSOURCE2 ;
PROC IML WORKSIZE=1000 SYMSIZE=1000;
&LINMOD ;
OPT_OFF = { MSH };    OPT_ON  = { LISTINFO AVAILOPT };
RUN SETOPT;
*Read raw data into IML;
USE PAYNE;
READ ALL VAR{GROUP} INTO GROUP;
READ ALL VAR{INIT}  INTO INIT;
READ ALL VAR{SCORE2 SCORE4 SCORE6 SCORE8 SCORE10} INTO Y;
CLOSE PAYNE;
*Use functions to create indicators for design matrix;
N=NROW(Y); * # observations in sample;
CONSTANT=J(N,1,1); *Column of 1's for intercept, etc;
CELLMEAN=DESIGN(GROUP);          *Cell mean coding;
EFFECT  =CONSTANT||DESIGNF(GROUP); *Effect coding;
REFERENC=CONSTANT||CELLMEAN(|*,2:NCOL(CELLMEAN)|); *Reference cell coding;
*Assemble all predictors and responses into one matrix, Z;
Z = CELLMEAN || INIT || Y;
ZNAMES = { CONTROL  LOW  MODERATE  HIGH  INIT }
          || { SCORE2 SCORE4 SCORE6 SCORE8 SCORE10 } ; * ZNAMES must also exist;
RUN MAKES; *Create SSCP matrix and associated parameters;
*Fit a model;
INDVARS = { CONTROL LOW MODERATE HIGH INIT };
DEPVARS = { SCORE2 SCORE4 SCORE6 SCORE8 SCORE10 };
RUN FITMODEL;
*Conduct any test or estimation desired;
*C= ---- ; *U= ---- ; *RUN TESTGLH;

```

Matrices Available from FITMODEL

MATRIX	OPTION	ROWNAME	COLNAME
<u>_BETA_</u>	BETA, EXBETA	<u>_XNAME_</u>	<u>_XNAME_</u>
<u>_BPVAL_</u>	EXBETA	<u>_XNAME_</u>	<u>_YNAME_</u>
<u>_BSE_</u>	EXBETA	<u>_XNAME_</u>	<u>_YNAME_</u>
<u>_BT_</u>	EXBETA	<u>_XNAME_</u>	<u>_YNAME_</u>
<u>_LINDEP_</u>	LINDEP	<u>_XNAME_</u>	<u>_VNAME_</u>
<u>_PARM1_</u>	UNIBETA	----	<u>_PM1CNM_</u>
<u>_SCORR_</u>	SCORR	<u>_YNAME_</u>	<u>_YNAME_</u>
<u>_SIGMA_</u>	SIGMA	<u>_YNAME_</u>	<u>_YNAME_</u>
<u>_SS_</u>	SSFIT	<u>_VNAME_</u>	<u>_VNAME_</u>
<u>_STAT_</u>	UNIBETA	<u>_YNAME_</u>	<u>_STRNM_</u>
<u>_XPXINV_</u>	XPXINV	<u>_XNAME_</u>	<u>_XNAME_</u>

Matrices Available from TESTGLH

Matrix	Option	Rowname	Colname	Section
<u>_THETA_</u>	THETA	<u>_THRNM_</u>	<u>_THCNM_</u>	7.2.2, 7.5.4
<u>_MID_</u>	MID	<u>_THRNM_</u>	<u>_THRNM_</u>	7.2.2, 7.5.5
<u>_SDTHTA_</u>	MATTHETA	<u>_THRNM_</u>	<u>_THCNM_</u>	7.5.7
<u>_TTHTA_</u>	MATTHETA	<u>_THRNM_</u>	<u>_THCNM_</u>	7.5.7
<u>_PVTHTA_</u>	MATTHETA	<u>_THRNM_</u>	<u>_THCNM_</u>	7.5.7
<u>_MSH_</u>	MSH	<u>_THCNM_</u>	<u>_THCNM_</u>	7.2.2, 7.5.9
<u>_MSE_</u>	MSE	<u>_THCNM_</u>	<u>_THCNM_</u>	7.2.2, 7.5.10
<u>_ECORR_</u>	ECORR	<u>_THCNM_</u>	<u>_THCNM_</u>	7.5.11
<u>_HEIVAL_</u>	HEIVAL	<u>_CANNM_</u>	<u>_NONM_</u>	7.5.12
<u>_CANVEC_</u>	CANVEC	<u>_THCNM_</u>	<u>_CANNM_</u>	7.5.13
<u>_CANRSQ_</u>	CANRSQ	<u>_CANNM_</u>	<u>_NONM_</u>	7.5.14
<u>_FSTATS_</u>	UNITHETA	<u>_THCNM_</u>	<u>_FSTRNM_</u>	7.5.6, 7.5.17
<u>_STMAT1_</u>	MULTTEST	<u>_STMRNM_</u>	<u>_STMCNM_</u>	7.5.16, 7.5.17
<u>_TPARM1_</u>	MULTTEST	<u>_NONM_</u>	<u>_TPCNM1_</u>	7.5.16
<u>_URESUL_</u>	UNIREP	<u>_UCOLNM_</u>	<u>_UROWNM_</u>	7.5.18-7.5.21

Note that if THETARNM exists then _THTRNM_=THETARNM.

Also, if THETACNM exists then _THTCNM_=THETACNM.

NAMELIST Function

Usage: yourlist = NAMELIST(STEM,LOW,HIGH,BY);

This function generates a row of names, STEM_{low} to STEM_{high}, by..., a character matrix. STEM is a character string (1x1). Require $0 \leq \text{LOW} \leq \text{HIGH}$, and $1 \leq \text{BY}$, integers (1x1). See the MAKESS example in section 4.9 of the full LINMOD manual for an example use. Example 1 in Section 1.5 of the full LINMOD manual includes the following code: DEPVAR=NAMELIST("SCORE",2,10,2);

UMEAN Function

Usage: UAVE = UMEAN(P); This function returns J(P,1,1/P), which provides a column (vector) to use as an averaging U matrix.

UPOLY1 Module

This module produces a U matrix and associated names for a test of trends.

Usage: RUN UPOLY1(VALUE, NAME, U, NMOUT);

Inputs. VALUES, numeric treatment levels (values), a matrix with one row or column.

NAME, a character string providing stem of names for trends.

Outputs. U, a matrix with columns orthonormal polynomial coefficients (excludes zero order).

NMOUT, a character matrix with one row of names.

UPOLY2 Module

This module produces U matrices and associated names for tests of trend and interaction for a design with two within-subject factors.

Assume Factor 1, with levels VALUES1, varies slowly,

and that Factor 2, with levels VALUES2, varies rapidly.

Usage:

```
RUN UPOLY2(VALUE1, NAME1, VALUE2, NAME2,
           U1, NMOUT1, U2, NMOUT2, U12, NMOUT12);
```

Inputs

- VALUES1=1st set of numeric treatment levels/values
- NAME1 = 1st character string providing stem of names;
- VALUES2=2nd set of numeric treatment levels/values
- NAME2 = 2nd character string providing stem of names;

Outputs

- U1 = orthonormal polynomial coefficient columns for 1st factor (excludes zero order)
- NMOUT1 = 1 row matrix of 1st factor names (character);
- U2 = orthonormal polynomial coefficient columns for 2nd factor (excludes zero order)
- NMOUT2 = 1 row matrix of 2nd factor names (character);
- U12 = orthonormal polynomial coefficient columns for interaction (excludes zero order)
- NMOUT12 = 1 row matrix of interaction names (character);

UTREND Function

Create polynomial trends matrix (columnwise), excluding the zero order.

Usage: UPOLY = UTREND(VALUE);